Constructing The Ideal Computer Game

COMPUTE!

$2.50
July
1983
Issue 38
Vol. 5, No. 7
£1.85 UK $3.25 Canada
63379
ISSN 0194-357X

The Leading Magazine Of Home, Educational, And Recreational Computing

Special Games Issue

Some Of The Finest Games Ever For VIC-20, 64, Atari, And Other Computers: *Roadblock, Castle Quest, Goblins,* And More!

Circles: A Machine Language Tutorial For Atari

Backing Up Your VIC-20 And 64 Disks

REM Revealed: A Tutorial For PET, VIC-20, And 64

PLUS:
Build Your Own Apple Data Manager, Stars On The Radio Shack Color Computer, Gold Miner Game For The TI-99/4A

Major Feature: New Products At The COMDEX Dealer Show

0 74470 63379 07

GOBLIN

Dan Goff

In "Goblin" (for the unexpanded VIC, 64, Atari, TI, and Apple) custom characters are used to create a simple yet entertaining game. The object is to capture the scowling creatures with your goblin while avoiding the many block-shaped obstacles that lie in your path.

After obstacles and sad faces have been positioned, "Goblin" begins when the main character appears at the bottom of the screen. As the game progresses, the goblin moves continually upward and the player controls only its horizontal movement. The "O" and "P" keys, in conjunction with the GET command in line 260, enable the player to move the goblin left and right, respectively. Children especially like the cumulative effect of the GET statement; they make rapid key punches and then wait for the delayed effects.

As each sad face is captured by the goblin, the score is updated and printed at the upper left. If the goblin successfully clears the screen of all the faces, an entirely new playfield will be provided. A game lasts as long as you wish.

A single round ends when the goblin crashes into an obstacle. At this point, the remaining sad faces smile, and you are asked if you wish to play again. If you don't, it is probably best to respond by typing "N" so that full memory is restored to the VIC.

On the other hand, if you play again, your previous highest score will be posted as the new game begins. The incentive to exceed a record score makes any game more fun.

64, AtARI, TI-99/4A And Apple Version Notes

The 64, Atari, TI-99/4A, and Apple versions of Goblin are almost identical to the VIC version. Minor differences do exist, however, in the Atari and Apple versions.

The Atari version uses the "+" and "*" keys to control left and right movement of the goblin. The Apple uses the left and right arrow keys.

The Apple version requires that you have a disk drive with the DOS Tool Kit disk in the drive when the program is run. This version defines certain characters using the program "Animatrix" from this disk. As Goblin is run, these custom characters are placed in memory as shapes and are later drawn on the high-resolution graphics screen. When the game begins, they are simultaneously POKEd into the areas of memory associated with the text and the high-resolution graphics screens. So, although you see these redefined characters on the high-resolution page, collision detection is actually carried out by PEEKing text screen memory.



Chasing goblins on the VIC-20 version of Goblin.

Goblin on the Commodore 64.

BEGINNING PROGRAMMERS

If you're new to computing, please read "How To Type COMPUTE!'s Programs" and "A Beginner's Guide To Typing In Programs."

Program 1: Goblin – VIC Version

```
100 PRINT"{CLR}":POKE 52,28:POKE 56,28:C
    LR:POKE 36869,255:POKE 36879,26
110 IFS>HSTHENHS=S
115 RESTORE:B=230:Z=8152:Z1=Z+30720:W=0:
    S=J:G=0
120 FOR X=1TO32:READ A:POKEX+7167,A:NEXT
    :FORX=1TO8:READA:POKEX+7423,A:NEXT
130 PRINT"{CLR}{RVS}{GRN}{5 RIGHT}G O B
    L I N"
140 PRINT"{HOME}{RED}{2 DOWN}"SPC(12)"
    {RVS}HS="HS:PRINT"{HOME}{RVS}{BLK}
    {20 DOWN}O=LEFT{9 RIGHT}P=RIGHT"
150 FOR I=1 TO 65
160 X=INT(RND(1)*330)+7746
170 IFPEEK(X)=BTHEN 160
180 POKEX,B:POKEX+30720,0:NEXTI
190 FORI=1TO20
200 X=INT(RND(1)*330)+7746
210 IF PEEK(X)=BORPEEK(X)=1ORPEEK(X)=3TH
    EN 200
220 IFPEEK(X+21)=BANDPEEK(X+22)=BANDPEEK
    (X+23)=BTHENPOKEX,3:POKEX+30720,0:G=
    G+1:GOTO240
230 POKEX,1:POKEX+30720,0
240 NEXT I
250 POKEZ,32:Z=Z-22:Z1=Z1-22:IF Z<7746 T
    HEN Z=Z+374:Z1=Z1+374
260 GET A$:IFA$="O"THENZ=Z-1:Z1=Z1-1
270 IFA$="P"THENZ=Z+1:Z1=Z1+1
280 IFPEEK(Z)=B THEN 410
290 IFPEEK(Z)=1 THEN GOSUB 330
300 POKEZ,0:POKEZ1,0:FORT=1TO220:NEXT
310 IFW=20-G THEN J=S:GOSUB350:GOTO110
320 GOTO 250
330 W=W+1:S=S+25:PRINT"{HOME}{BLU}
    {2 DOWN}{RVS}"S:POKE36878,15
340 FORT=235TO250:POKE36876,T:NEXT:POKE3
    6876,0:RETURN
350 PRINT"{HOME}{RED}{16 DOWN}{RVS}*****
    *ALL RIGHT!******"
355 FORI=1TO10:GETA$:NEXTI:REM COLLECT G
    ARBAGE
360 FORI=1TO25
370 X=INT(RND(1)*15)+233
380 POKE36878,15:POKE36875,X
390 FORT=1TO30:NEXTT:NEXTI
400 POKE36878,0:POKE36875,0:RETURN
410 POKE36877,200:FORV=15TO0STEP-1:POKE3
    6878,V:NEXT:POKE36877,0:POKEZ,2
420 FORX=7746TO8075:IF PEEK(X)<>1THEN NE
    XTX
430 IFPEEK(X)=1THEN POKEX,3:NEXTX
440 J=0
445 FORI=1TO10:GET C$:NEXTI
450 PRINT"{HOME}{BLU}{18 DOWN}{RIGHT}
    {RVS}PLAY AGAIN? (Y/N)"
465 GET C$:IF C$="" THEN 465
470 IFC$="Y"THEN 110
490 POKE 36869,240:POKE36879,27:POKE52,3
    0:POKE56,30:PRINT"{CLR}SEE YA!"
500 DATA126,219,219,255,165,90,90,165,60
    ,66,165,129,153,165,66,60
510 DATA 170,85,170,85,126,219,255,189,6
    0,66,165,129,165,153,66,60
520 DATA 0,0,0,0,0,0,0,0
```

Program 2: Goblin – 64 Version

```
80 POKE 53280,2:POKE 53281,1
90 PRINT"{CLR}{7 DOWN}{4 RIGHT}PLEASE WA
   IT...DEFINING CHARACTERS";
100 POKE 52,48:POKE 56,48:CLR:POKE56334,
    PEEK(56334)AND254
105 POKE1,PEEK(1)AND251
108 FORN=0TO2047:POKEN+12288,PEEK(N+5324
    8):NEXTN
109 FOR N=0 TO 7:POKEN+12320,PEEK(N+5406
    4):NEXT N
110 IFS>HSTHENHS=S
112 RESTORE:B=4:Z=1964:Z1=Z+54272:W=0:S=
    J:G=0
115 VS=54296:AD=54277:SR=54278:WF=54276:
    LB=54272:HB=54273
120 FOR X=0TO31:READ A:POKEX+12288,A:NEX
    T
123 POKE 1,PEEK(1)OR4:POKE56334,PEEK(563
    34)OR1
125 POKE 53272,(PEEK(53272)AND240)+12
130 PRINT"{CLR}{GRN}{14 RIGHT}{RVS}G O B
     L I N"
140 PRINT"{HOME}{RED}{2 DOWN}{RVS}"SPC(1
    7)"HS="HS
145 PRINT"{HOME}{BLK}{22 DOWN}{RVS}O=LEF
    T";SPC(27);"P=RIGHT"
150 FOR I=1 TO 118
160 X=INT(RND(1)*680)+1144
170 IFPEEK(X)=BTHEN 160
180 POKEX,B:POKEX+54272,0:NEXTI
190 FORI=1TO36
195 G1=0
200 X=INT(RND(1)*680)+1144
210 IF PEEK(X)=BORPEEK(X)=1ORPEEK(X)=3TH
    EN 200
220 IFPEEK(X+39)=BANDPEEK(X+40)=BANDPEEK
    (X+41)=BTHENPOKEX,3:POKEX+54272,0:G1
    =1
225 IF G1=1 THEN G=G+1:GOTO 240
230 POKEX,1:POKEX+54272,0
240 NEXT I
250 POKEZ,32:Z=Z-40:Z1=Z1-40:IF Z<1144 T
    HEN Z=Z+760:Z1=Z1+760
260 GET A$:IFA$="O"THENZ=Z-1:Z1=Z1-1
270 IFA$="P"THENZ=Z+1:Z1=Z1+1
280 IFPEEK(Z)=B THEN 410
290 IFPEEK(Z)=1 THEN GOSUB 330
300 POKEZ,0:POKEZ1,0:FORT=1TO220:NEXT
310 IFW=36-G THEN J=S:GOSUB350:GOTO110
320 GOTO 250
330 W=W+1:S=S+25:PRINT"{HOME}{BLU}
    {2 DOWN}"S:POKE VS,15:POKE AD,30:POK
    E SR,200:POKE WF,17
340 POKEHB,71:POKELB,12:FORT=1TO90:NEXTT
    :POKEVS,0:POKEHB,0:POKELB,0:RETURN
350 PRINT"{HOME}{RED}{18 DOWN}{8 RIGHT}
    {RVS}******ALL RIGHT!******"
355 FORI=1TO10:GETC$:NEXTI:REM COLLECT G
    ARBAGE
360 POKE VS,15:POKE AD,30:POKE SR,200:PO
    KE WF,17:FOR I=1 TO 17
370 H=INT(RND(0)*10)+21:L=INT(RND(0)*45)
    +210:POKE HB,H:POKE LB,L
380 FOR T=1 TO 80:NEXT T:NEXTI:POKE VS,0
    :POKE HB,0:POKE LB,0
```

```
400 RETURN
410 POKE2,2:POKEVS,15:POKEAD,30:POKESR,2
    00:POKEWF,129:POKE HB,2:POKE LB,125
415 FOR I=1 TO 400:NEXT I:POKE VS,15:POK
    E HB,0:POKE LB,0
420 FORX=1144TO1823:IF PEEK(X)<>1THEN NE
    XTX
430 IFPEEK(X)=1THEN POKEX,3:NEXTX
440 J=0
445 FORI=1TO10:GET C$:NEXTI
450 PRINT"{HOME}{BLU}{20 DOWN}{RVS}PLAY
    AGAIN? (Y/N)":POKE 646,14
465 GET C$:IF C$="" THEN 465
470 IFC$="Y"THEN 110
490 POKE53272,21:POKE53280,14:POKE53281,
    6:POKE 52,50:POKE56,50:PRINT"{CLR}SE
    E YA!"
500 DATA126,219,219,255,165,90,90,165,60
    ,66,165,129,153,165,66,60
510 DATA 170,85,170,85,126,219,255,189,6
    0,66,165,129,165,153,66,60
520 DATA 0,0,0,0,0,0,0,0
```

Program 3: Goblin – Atari Version

```
90 SCREEN=PEEK(88)+256*PEEK(89):DIM
   A$(3):OPEN #1,4,0,"K:"
100 GRAPHICS 1+16:POSITION 1,10:? #6
    ;"...PLEASE WAIT..."
105 GOSUB 2000
107 IF S>HS THEN HS=S
108 S=J1:W=0:Z=SCREEN+900:G=0
110 GRAPHICS 0:POKE 752,1:SETCOLOR 4
    ,6,6:SETCOLOR 2,10,1:POSITION 13
    ,0:PRINT "G O B L I N"
115 POKE 756,CHSET/256
120 POSITION 16,2:? "HS :";HS:POSITI
    ON 1,22:? "LEFT=+ KEY":POSITION
    28,22:? "RIGHT=* KEY";
150 FOR I=1 TO 120
160 X=SCREEN+INT(RND(0)*640)+160
170 IF PEEK(X)=7 THEN 160
180 POKE X,7:NEXT I
190 FOR I=1 TO 36
200 X=SCREEN+INT(RND(0)*640)+160
210 IF PEEK(X)=7 OR PEEK(X)=1 OR PEE
    K(X)=32 THEN 200
220 IF PEEK(X+39)=7 AND PEEK(X+40)=7
     AND PEEK(X+41)=7 THEN POKE X,1:
    G=G+1:GOTO 240
230 POKE X,32
240 NEXT I
245 SOUND 1,50,10,12:FOR I=1 TO 50:N
    EXT I:SOUND 1,0,0,0:FOR I=1 TO 2
    00:NEXT I
250 POKE Z,0:Z=Z-40:IF Z<SCREEN+120
    THEN Z=Z+760
260 A=PEEK(764):POKE 764,255:IF A=7
    THEN Z=Z+1
270 IF A=6 THEN Z=Z-1
280 IF PEEK(Z)=7 THEN 410
290 IF PEEK(Z)=32 THEN GOSUB 330
300 POKE Z,5:FOR T=1 TO 100:NEXT T
310 IF W=36-G THEN J1=S:GOSUB 350:GO
    TO 107
320 GOTO 250
330 W=W+1:S=S+25:POSITION 3,2:? S
340 SOUND 2,20,14,12:FOR I=1 TO 20:S
    OUND 2,0,0,0
345 RETURN
350 FOR I=SCREEN+360 TO SCREEN+480:P
    OKE I,0:NEXT I:POSITION 10,10:?
    "**** ALL RIGHT ***"
355 J1=S
360 FOR X=1 TO 20:SOUND 1,30-X,10,12
    :FOR I=1 TO 40:NEXT I:NEXT X:SOU
    ND 1,0,0,0
400 RETURN
410 POKE Z,6
415 FOR V=12 TO 0 STEP -1:SOUND 1,40
    ,2,V:SOUND 2,70,12,V:SETCOLOR 4,
    V,6:FOR I=1 TO 40:NEXT I:NEXT V
418 SETCOLOR 4,6,6:SOUND 1,0,0,0:SOU
    ND 2,0,0,0
420 FOR X=SCREEN+160 TO SCREEN+800:I
    F PEEK(X)<>32 THEN NEXT X
430 IF PEEK(X)=32 THEN POKE X,1:NEXT
     X
440 J1=0
450 POKE 764,255:POSITION 10,21:? "P
    lay Again (Y/N)?";:GET #1,A
460 IF A=ASC("Y") THEN 107
470 GRAPHICS 1+16:POSITION 3,10:? #6
    ;"...SEE YA..":FOR I=1 TO 800:N
    EXT I:END
2000 CHSET=(PEEK(106)-8)*256:FOR I=0
      TO 1023:POKE CHSET+I,PEEK(5734
    4+I):NEXT I
2001 RESTORE 2005
2002 READ A:IF A=-1 THEN RETURN
2003 FOR J=0 TO 7:READ B:POKE CHSET+
    A*8+J,B:NEXT J
2004 GOTO 2002
2005 DATA 1,60,126,219,255,189,195,1
    26,60
2006 DATA 5,60,126,219,255,195,153,2
    55,255
2007 DATA 6,204,204,51,51,204,126,21
    9,255
2008 DATA 7,204,204,51,51,204,204,51
    ,51
2009 DATA 32,60,126,219,255,231,219,
    126,0
2010 DATA -1
```



Atari version of Goblin.

Hello, We're Adventure International



And we're publishers of some of the finest microcomputer software programs available.
If you can write a **top-quality** program, or can convert some of our best-sellers to other computers, we want to hear from you — Now.
We have the advertising, international distribution, manufacturing and marketing know-how to send top-quality programs to the top of the charts.
If your program **is top quality** — give us a call, or write for our Adventure International Author Information Kit.

We are publishers of the top-selling Scott Adams Adventure Series and other fine Entertainment and Applications Programs.

Adventure INTERNATIONAL

Box 3435
Longwood, Florida 32750
Telephone: (305) 862-6917
Ask for Author Assistance

Program 4: Goblin – TI-99/4A Version

```
100 RANDOMIZE
110 GOTO 170
120 FOR I=1 TO LEN(H$)
130 R=ASC(SEG$(H$,I,1))
140 CALL HCHAR(ROW,XCOL+I,R)
150 NEXT I
160 RETURN
170 A=96
180 B=97
190 C=104
200 D=105
210 Z=24
220 COL=16
230 W=0
240 G=0
250 S=J
260 CALL CLEAR
270 IF S>HS THEN 290
280 GOTO 300
290 HS=S
300 GOSUB 1270
310 CALL SCREEN(16)
320 PRINT "{8 SPACES}G O B L I N"
330 PRINT
340 PRINT "{10 SPACES}HS : "
350 FOR I=1 TO 19
360 PRINT
370 NEXT I
380 PRINT "O=LEFT{14 SPACES}P=RIGHT"
    ;
390 ROW=4
400 XCOL=17
410 H$=STR$(HS)
420 GOSUB 120
430 FOR I=1 TO 80
440 X=INT(RND*30)+2
450 Y=INT(RND*16)+6
460 CALL GCHAR(Y,X,L)
470 IF L=B THEN 440
480 CALL HCHAR(Y,X,B)
490 NEXT I
500 FOR I=1 TO 27
510 X=INT(RND*30)+2
520 Y=INT(RND*16)+6
530 CALL GCHAR(Y,X,L)
540 IF (L=B)+(L=C)+(L=D)THEN 510
550 CALL GCHAR(Y+1,X-1,L)
560 CALL GCHAR(Y+1,X,M)
570 CALL GCHAR(Y+1,X+1,N)
580 IF (L<>B)+(M<>B)+(N<>B)THEN 620
590 CALL HCHAR(Y,X,D)
600 G=G+1
610 GOTO 630
620 CALL HCHAR(Y,X,C)
630 NEXT I
640 CALL SOUND(100,500,6)
650 CALL HCHAR(Z,COL,32)
660 IF L<>C THEN 680
670 CALL SOUND(10,880,4)
680 Z=Z-1
690 IF Z>4 THEN 710
700 Z=23
710 CALL KEY(0,L,ST)
720 IF (L<>79)*(L<>80)THEN 770
730 IF L<>79 THEN 760
740 COL=COL-SGN(COL-2)
750 GOTO 770
760 COL=COL+SGN(30-COL)
770 CALL GCHAR(Z,COL,L)
780 IF L=B THEN 1060
790 IF L=C THEN 850
800 CALL HCHAR(Z,COL,A)
810 FOR I=1 TO 25
820 NEXT I
830 IF W=27-G THEN 920
840 GOTO 650
850 W=W+1
860 S=S+25
870 H$=STR$(S)
880 ROW=4
890 XCOL=3
900 GOSUB 120
910 GOTO 800
920 J=S
930 CALL HCHAR(10,1,32,31)
940 GOSUB 120
950 H$="**** ALL RIGHT! ****"
960 XCOL=6
970 ROW=10
980 GOSUB 120
990 FOR I=1 TO 15
1000 X=INT(RND*100)+300
1010 CALL SOUND(75,X,8)
1020 NEXT I
1030 FOR I=1 TO 100
1040 NEXT I
1050 GOTO 210
1060 REM WHOOPS! ...YOU CRASHED...
1070 CALL HCHAR(Z,COL,98)
1080 FOR I=3 TO 30 STEP 3
1090 CALL SOUND(50,-7,I)
1100 NEXT I
1110 CALL CHAR(104,"3C42A581A599423
     C")
1120 J=0
1130 HS=S
1140 H$="PLAY AGAIN (Y / N)?"
1150 ROW=22
1160 XCOL=2
1170 GOSUB 120
1180 CALL KEY(0,L,ST)
1190 IF ST=0 THEN 1180
1200 H$=CHR$(L)
1210 IF H$="Y" THEN 1250
1220 CALL CLEAR
1230 PRINT "SEE YA!"
1240 END
1250 CALL CHAR(104,"3C3CA58199A5423
     C")
1260 GOTO 210
1270 REM DEFINE CUSTOM CHARS
1280 REM CHAR 96 - GOBLIN
1290 CALL CHAR(96,"7EDBDBFFA55A5AA5
     ")
1300 REM CHAR 97 - BARRIER
1310 CALL CHAR(97,"CCCC3333CCCC3333
     ")
1320 REM CHAR 98 - CRUNCHED GOBLIN
1330 CALL CHAR(98,"CCCC33337EDBFFBD
     ")
1340 REM CHAR - 104 - FROWN
1350 CALL CHAR(104,"3C3CA58199A5423
     C")
1360 REM CHAR - 105 - SMILE
1370 CALL CHAR(105,"3C42A581A599423
     C")
1380 CALL COLOR(10,7,1)
1390 FOR I=5 TO 8
1400 CALL COLOR(I,16,14)
1410 NEXT I
1420 RETURN
```



Goblin, TI-99/4A version.



The Apple version of Goblin.

Program 5: Goblin – Apple Version

```
10  REM *THIS PROGRAM REQUIRES A DISK D
    RIVE
20  REM  AND THE APPLE 'DOS TOOL KIT PR
    OGRAMMING
30  REM  UTILITIES DISK' TO RUN.
40  REM *PLACE THE ABOVE UTILITY DISK I
    N YOUR
50  REM  DRIVE BEFORE RUNNING THIS PROG
    RAM.
55  GOSUB 1000
60  DIM XL%(23): FOR I = 0 TO 7:XL%(I) =
     1024 + 128 * I:XL%(I + 8) = 1064 +
    128 * I:XL%(I + 16) = 1104 + 128 *
    I: NEXT I
85  HOME : HGR : POKE  - 16302,0: IF S >
    HS THEN HS = S
90 ZROW = 23:ZCOL = 19:W = 0:S = J1:G =
    0
100  VTAB 1: HTAB 17: PRINT  CHR$ (1);"
    O"; CHR$ (9);"G O B L I N";
110  VTAB 2: HTAB 19: PRINT "HS"; CHR$
    (14);": ";HS: VTAB 23: PRINT "LEFT
    ="; CHR$ (9);"LEFT ARROW"; CHR$ (1
    4);
120  HTAB 24: PRINT "RIGHT="; CHR$ (9);
    "RIGHT ARROW"; CHR$ (14);
125  PRINT  CHR$ (1);"1";
130  FOR I = 1 TO 120
135 ROW =  INT ( RND (1) * 15) + 5:COL =
     INT ( RND (1) * 38) + 2
140 X = XL%(ROW) + COL: IF  PEEK (X) =
    164 THEN 135
145  VTAB ROW: HTAB COL: PRINT  CHR$ (1
    64);: NEXT I
150  FOR I = 1 TO 36
160 ROW =  INT ( RND (1) * 15) + 5:COL =
     INT ( RND (1) * 38) + 2:X = XL%(R
    OW - 1) + COL - 1
170  IF  PEEK (X) = 164 OR  PEEK (X) =
    161 OR  PEEK (X) = 163 THEN 160
180  IF  PEEK (XL%(ROW) + COL - 2) = 16
    4 AND  PEEK (XL%(ROW) + COL - 1) =
    164 AND  PEEK (XL%(ROW) + COL) = 1
    64 THEN  HTAB COL: VTAB ROW: PRINT
     CHR$ (161);:G = G + 1: GOTO 200
190  HTAB COL: VTAB ROW: PRINT  CHR$ (163);
200  NEXT I
240  POKE 768,5: POKE 769,180: CALL 770
250 Z = XL%(ZROW) + ZCOL: HTAB ZCOL: VTAB
    ZROW: PRINT  CHR$ (167);:ZROW = ZR
    OW - 1: IF ZROW < 3 THEN ZROW = 21
260 A =  PEEK ( - 16384): POKE  - 16368
    ,0: IF A = 136 THEN ZCOL = ZCOL -
    1: IF ZCOL < 1 THEN ZCOL = 39
270  IF A = 149 THEN ZCOL = ZCOL + 1: IF
    ZCOL > 39 THEN ZCOL = 2
280  IF  PEEK (XL%(ZROW - 1) + ZCOL - 1
    ) = 164 THEN 410
290  IF  PEEK (XL%(ZROW - 1) + ZCOL - 1
    ) = 163 THEN  GOSUB 330
300  HTAB ZCOL: VTAB ZROW: PRINT  CHR$
    (165);: FOR T = 1 TO 100: NEXT T
310  IF W = 36 - G THEN J = S: GOSUB 35
    0: GOTO 85
320  GOTO 250
330 W = W + 1:S = S + 25: VTAB 2: HTAB
    3: PRINT  CHR$ (1);"O";S; CHR$ (14
    ); CHR$ (1);"1";
340  POKE 768,2: POKE 769,230: CALL 770
345  RETURN
350  FOR J = 10 TO 12: VTAB J: FOR I =
    0 TO 39: HTAB I: PRINT  CHR$ (167)
    ;: NEXT I: NEXT J: VTAB 17: HTAB 1
    0: PRINT  CHR$ (1);"O";"****** ALL
     RIGHT! ******"; CHR$ (1);"1";
360  FOR I = 1 TO 10
370  POKE 768, INT ( RND (1) * 3) + 1: POKE
    769, INT ( RND (1) * 15) + 130: CALL
    770
380  NEXT I
385 J1 = S
390  FOR J = 1 TO 500: NEXT J
400  RETURN
410  HTAB ZCOL: VTAB ZROW: PRINT  CHR$
    (166);:C = 0
415 X =  PEEK ( - 16336):C = C + 1: IF
    C < 15 THEN 415
420  FOR ROW = 0 TO 23: FOR COL = 1 TO
    38:X = XL%(ROW) + COL: IF  PEEK (X
    ) <  > 163 THEN  NEXT COL: NEXT ROW
430  IF  PEEK (X) = 163 THEN  VTAB ROW +
    1: HTAB COL + 1: PRINT  CHR$ (161)
    ;: NEXT COL: NEXT ROW
440 J1 = 0: VTAB 21: HTAB 13: PRINT  CHR$
    (1);"O";"PLAY AGAIN ("; CHR$ (9);"
```

```
    Y"; CHR$ (14);"/"; CHR$ (9);"N"; CHR$
    (14);") ?";
450 POKE  - 16368,0: GET C$: IF C$ = "
    Y" THEN 85
460 TEXT : HOME : VTAB 4: HTAB 2: PRINT
    "SEE YA!...HIT RESET...": FOR I =
    1 TO 1000: NEXT I: END
1000  REM  INIT SUBROUTINE
1020 ADRS = 0
1030  PRINT  CHR$ (4);"BLOAD RBOOT"
1040  CALL 520
1050 ADRS =  USR (0),"HRCG"
1051  IF ADRS < 0 THEN ADRS = ADRS + 65536
1060 CS = ADRS - 768
1061 CH =  INT (CS / 256):CL = CS - CH *256
1062  POKE ADRS + 7,CL: POKE ADRS + 8,CH
1070  HIMEM: CS
1080  READ A: IF A =  - 1 THEN 1100
1090  FOR J = 0 TO 7: READ B: POKE CS +
    A * 8 + J,B: NEXT : GOTO 1080
1100  CALL ADRS + 3
1110  PRINT  CHR$ (16)
1150  REM  SOUND ROUTINE
1160  FOR I = 770 TO 795: READ M: POKE
    I,M: NEXT
1170  RETURN
1500  DATA  1,28,62,127,107,127,93,34,28
1510  DATA  3,28,62,107,127,99,93,62,28
1520  DATA  4,85,42,85,42,85,42,85,42
1530  DATA  5,28,62,107,127,107,85,127,119
1540  DATA  6,85,42,85,42,85,62,107,127
1545  DATA  7,0,0,0,0,0,0,0,0
1550  DATA  -1
1560  DATA  172,1,3,174,1,3,169,4,32,16
    8,252,173,48,192,232,208,253,136,2
    08,239,206,0,3,208,231,96
```

©